


On souhaite gérer un match de volley-ball en temps réel. L'objectif est de faire la somme des points marqués et des sets remportés au fur et à mesure du déroulement de la partie.

- 1) Créer une application conforme au modèle ci-contre. Indiquer votre nom et votre prénom directement sur la feuille. Enregistrer le projet sur votre disquette en utilisant votre nom de famille comme nom des fichiers. Ne pas oublier de donner des noms explicites aux composants visuels qui seront utilisés dans le code. [5 points]
- 2) Saisir la ligne d'instruction « Option Explicit » dans la partie « (Général) (Déclarations) » du code afin d'éviter toute faute de frappe lors de la saisie du nom d'une variable existante ou d'écartier tout risque de confusion dans un code où la portée de la variable n'apparaît pas clairement.
- 3) Prévoir les variables globales suivantes : scores des deux équipes, nombre de sets remportés par chaque équipe, numéro du set en cours. [1 point]
- 4) Programmer les deux boutons « OK » de manière à faire apparaître le nom des équipes dans les étiquettes situées au dessus des listes rappelant le score de chaque équipe. [1 point]
- 5) Programmer le bouton « Nouveau match » de manière à vider les zones de texte, les listes d'édition et les étiquettes contenant le nom des équipes. Remettre également le choix du nombre de sets gagnants à deux. On pourra utiliser la méthode *Clear* des composants *ListBox* pour effacer simplement les listes. On pourra utiliser la procédure associée au bouton pour initialiser les variables en début de programme. [3 points]
- 6) Programmer les boutons « Point pour l'équipe x » de manière à gérer l'évolution des points de chaque équipe. Se référer à l'algorithme fourni en annexe. Le traitement est pratiquement identique pour les deux boutons : il est possible de le tester pour l'équipe 1 puis de copier/coller le code et de l'adapter pour l'équipe 2. Le numéro du set en cours et le score actuel doivent apparaître dans les étiquettes correspondantes. Utiliser la procédure prédéfinie *MsgBox** pour indiquer l'équipe gagnante. [8 points]
- 7) Programmer le bouton « Quitter » pour sortir de l'application. [1 point]
- 8) Prévoir un moyen d'empêcher la poursuite d'un match déjà terminé (i.e., ne pas ajouter de point à une équipe quand on clique sur le bouton correspondant si elle a déjà gagné – ou perdu). [1 point]



Annexe

Rappel : Un set se joue normalement en 15 points, mais il faut deux points d'écart avec l'adversaire pour le remporter. Normalement, la première équipe qui remporte deux sets gagne le match. Pour les besoins de l'examen, nous introduisons la possibilité de gagner un match en remportant trois sets.

Algorithme boutons « Points pour l'équipe x » (en français) :

Augmenter le score de l'équipe considérée d'1 point

Si le score de l'équipe est supérieur ou égal à 15 alors

On teste si l'équipe a gagné le set

Si la différence de points entre les deux équipes est supérieure ou égale à 2 alors

L'équipe a gagné le set !

Augmenter le nombre de sets gagnés par l'équipe de 1

Remettre les scores à 0

Augmenter le numéro du set en cours de 1

On teste si l'équipe a gagné le match

Si (le nombre de sets gagnants est 2 et le nombre de sets gagnés par l'équipe est 2)

ou (le nombre de sets gagnants est 3 et le nombre de sets gagnés par l'équipe est 3) alors

L'équipe a gagné le match !

Fin si

Fin si

Fin si

* EX. MsgBox ("Ce message va s'afficher")

Code

Option Explicit

```
Dim score1 As Integer ' score de l'équipe 1
Dim score2 As Integer ' score de l'équipe 2
Dim set1 As Integer ' nombre de sets remportés par l'équipe 1
Dim Set2 As Integer ' nombre de sets remportés par l'équipe 2
Dim no_set As Integer ' set courant
Dim fin As Boolean ' indicateur de fin de match
```

```
Private Sub Form_Load()
    Nouveau_Click
End Sub
```

```
Private Sub Nouveau_Click()
```

```
    score1 = 0
    set1 = 0
    score2 = 0
    Set2 = 0
    deux_sets.Value = True
    NomEq1.Text = ""
    NomEq2.Text = ""
    no_set = 1
    NoSet = "Set n° " + Str(no_set)
    Eq1.Caption = "Equipe 1"
    Eq2.Caption = "Equipe 2"
    Liste_sets.Clear
    Liste_scores1.Clear
    Liste_scores2.Clear
    fin = False
```

```
End Sub
```

```
Private Sub PtEq1_Click()
```

```
    If Not fin Then
        score1 = score1 + 1
        If score1 >= 15 Then ' gagné un set ?
            If (score1 - score2) >= 2 Then ' gagné un set !
                set1 = set1 + 1
                Liste_sets.AddItem (Str(no_set))
                Liste_scores1.AddItem (Str(score1))
                Liste_scores2.AddItem (Str(score2))
                score1 = 0
                score2 = 0
                no_set = no_set + 1
                NoSet.Caption = "Set n° " + Str(no_set)
                If (deux_sets.Value And set1 >= 2) Or (trois_sets.Value And set1 >= 3) Then
                    ' gagné le match
                    MsgBox ("Match terminé : " + NomEq1.Text + " vainqueur")
                    Beep
                    fin = True
                End If
            End If
        End If
        ' rafraichissement de l'affichage
        Score.Caption = "Score : " + Str(score1) + " - " + Str(score2)
    Else
        Beep
    End If
End Sub
```

```
Private Sub PtEq2_Click()
```

```
    If Not fin Then
        score2 = score2 + 1
        If score2 >= 15 Then ' gagné ?
            If (score2 - score1) >= 2 Then ' gagné !
                Set2 = Set2 + 1
                Liste_sets.AddItem (Str(no_set))
                Liste_scores1.AddItem (Str(score1))
                Liste_scores2.AddItem (Str(score2))
                score1 = 0
                score2 = 0
                no_set = no_set + 1
                NoSet.Caption = "Set n° " + Str(no_set)
                If (deux_sets.Value And Set2 >= 2) Or (trois_sets.Value And Set2 >= 3) Then
                    MsgBox ("Match terminé : " + NomEq2.Text + " vainqueur")
                    Beep
                    fin = True
                End If
            End If
        End If
        ' rafraichissement de l'affichage
        Score.Caption = "Score : " + Str(score1) + " - " + Str(score2)
    Else
        Beep
    End If
End Sub

Private Sub Quitter_Click()
    End
End Sub

Private Sub ValidEq1_Click()
    Eq1.Caption = NomEq1.Text
End Sub

Private Sub ValidEq2_Click()
    Eq2.Caption = NomEq2.Text
End Sub
```